@PACKAGE_INIT@
include(CMakeFindDependencyMacro)

set_and_check(prometheus-cpp_INCLUDE_DIR "@PACKAGE_CMAKE_INSTALL_INCLUDEDIR@")

set(PROMETHEUS_CPP_ENABLE_PULL @ENABLE_PULL@)
set(PROMETHEUS_CPP_ENABLE_PUSH @ENABLE_PUSH@)
set(PROMETHEUS_CPP_USE_COMPRESSION @ENABLE_COMPRESSION@)
set(PROMETHEUS_CPP_USE_THIRDPARTY_LIBRARIES @USE_THIRDPARTY_LIBRARIES@)
set(PROMETHEUS_CPP_THIRDPARTY_CIVETWEB_WITH_SSL @THIRDPARTY_CIVETWEB_WITH_SSL@)

set(CMAKE_THREAD_PREFER_PTHREAD TRUE)
find_dependency(Threads)
unset(CMAKE_THREAD_PREFER_PTHREAD)

if(PROMETHEUS_CPP_ENABLE_PULL)
  if(PROMETHEUS_CPP_USE_THIRDPARTY_LIBRARIES)
    if(PROMETHEUS_CPP_THIRDPARTY_CIVETWEB_WITH_SSL)
        find_dependency(OpenSSL)
    endif()
  else()
    find_dependency(civetweb)
  endif()
endif()

if(PROMETHEUS_CPP_ENABLE_PULL AND PROMETHEUS_CPP_USE_COMPRESSION)
  find_dependency(ZLIB)
endif()

if(PROMETHEUS_CPP_ENABLE_PUSH)
  find_dependency(CURL)
endif()

include("${CMAKE_CURRENT_LIST_DIR}/prometheus-cpp-targets.cmake")
